

Physics

HP COMPUTER CURRICULUM

Mechanics

TEACHERS ADVISOR

HEWLETT  PACKARD

Hewlett-Packard
Computer Curriculum Series

physics
TEACHER'S ADVISOR

mechanics

by Herbert D. Peckham
Gavilan College

edited by Briana Burns and Jean Danver
Hewlett-Packard

Hewlett-Packard Company
11000 Wolfe Road
Cupertino, California 95014

© Copyright, 1972, by
HEWLETT-PACKARD COMPANY
Cupertino, California
Printed in the U.S.A.

The Hewlett-Packard Computer Curriculum Series represents the published results of a Curriculum Development Project sponsored by the Data Systems Division of Hewlett-Packard Company. This project is under the directorship of Jean H. Danver.

This material is designed to be used with any Hewlett-Packard system with the BASIC programming language such as the 9830A, Educational BASIC, and the 2000 and 3000 series systems.

The Hewlett-Packard Company makes no warranty expressed or implied, and assumes no responsibility in connection with the operation of the program material contained herein.

Copyright © 1972 by Hewlett-Packard Company, Cupertino, California. All rights reserved. No part of this publication may be reproduced, stored in a retrieval system (e.g., in memory, disc or core) or be transmitted by any means, electronic, mechanical, photocopy, recording or otherwise, without prior written permission from the publisher.

Printed in the U.S.A.

TABLE OF CONTENTS

1 INTRODUCTION

3 RATES

3 Exercise 1 — Computing Velocity

4 Exercise 2 — Applications

6 Exercise 3 — Automatic Accuracy

Advanced Exercises

7 Exercise 4 — Plotting Results

7 Exercise 5 — Rates and Trigonometric Functions

9 FROM RATES TO DISPLACEMENTS

9 Exercise 6 — Displacement

10 Exercise 7 — Increasing the Accuracy

11 Exercise 8 — Finding Displacement from Acceleration

Advanced Exercises

12 Exercise 9 — Average Values of Velocity & Acceleration

13 Exercise 10 — A Generalization

15 NEWTON'S SECOND LAW

15 Exercise 11 — Force and Motion

16 Exercise 12 — Converging on Accuracy

18 Exercise 13 — A Variable Force

Advanced Exercises

19 Exercise 14 — A Challenge

21 Exercise 15 — Barrier Penetration

23 HALF STEP METHOD

23 Exercise 16 — Half Step Computation

24 Exercise 17 — Half Step Computation

25 THE HARMONIC OSCILLATOR

25 Exercise 18 — Harmonic Oscillator

27 Exercise 19 — Half Step Computation

29 Exercise 20 — Changing the Mass

Advanced Exercise

31 Exercise 21 — The Pendulum

TABLE OF CONTENTS (Continued)

33 MORE COMPLICATED FORCES

33 Exercise 22 — Damped Harmonic Oscillator

35 Exercise 23 — An Experiment

Advanced Exercise

35 Exercise 24 — Driven, Damped Harmonic Oscillator

37 ORBITAL MOTION

37 Exercise 25 — Orbital Motion

38 Exercise 26 — Experimentation

38 Exercise 27 — A Challenge

Advanced Exercise

39 Exercise 28 — A Double Force Center

INTRODUCTION

This Physics Problem-Solving Set consists of a Student Lab Book and the corresponding Teacher's Advisor. It was designed to help meet the need for computer-oriented problems in physics, providing students an opportunity to use a computer as a problem solving tool within a particular subject matter area.

The materials are designed for flexible use as desired by the individual instructor. The material and exercises in this unit are intended as an "enrichment" experience in the field of mechanics.

Nearly all the exercises involve calculus in a difference approximation. This enables a wide variety of problems to be undertaken which would be impossible using only algebra and trigonometry. However, the words calculus, derivative, and integration are never used. The student who uses this material will have a much better feeling for calculus when he studies it analytically. Of course the primary reason a student should use the material is to better understand the ideas in mechanics. The use of this material will not compete with your text. Instead, it can be used to supplement and enrich in any fashion you choose.

The degree of difficulty of the material is dependent upon the amount of assistance which you choose to provide. With no assistance, the better physics student should be challenged. However, given a good deal of assistance, any physics student should be able to work out the exercises with no great difficulty. The level of the material is determined by the assumption that students taking introductory physics will be quite capable as a group.

The Lab Books provide text material and programming exercises for the students, advanced problems, and a sample program. The Teacher's Advisor contains an example of a program to solve each exercise and a brief discussion of the important elements of exercise.

For best results, you should study all the solutions until you are certain you have a complete grasp of the general methods. This should be done before assigning the material to the class. Generally, the exercises are cumulative so that as techniques are developed they are used in subsequent exercises. Therefore, you will probably wish to proceed through the exercises in the order in which they are given.

The solutions should be treated as *typical* only. There are many ways to work a given problem. Students should be encouraged to develop their own solutions even though they may be quite different from those presented here. Also, you should not become obsessively concerned with computer efficiency. It is far more important that the student develop a logical and methodical approach to computer programming than to dwell on flashy techniques.

NOTES

RATES

Exercise 1 — Computing Velocity

This exercise utilizes the program from Figure 1 in the Student Lab Book. The only change required is a new DEF statement in line 150. The intent is to let the student see the limiting process as the average and instantaneous velocities approach each other. If the student tries a value of D too small, the accuracy begins to deteriorate due to internal round off error by the computer. If you choose, you might ask the student to investigate other values of time and plot the results.

RUN

```

100 REM PROGRAM FOR AVERAGE VELOCITY
110 PRINT "INPUT VALUE OF T DESIRED";
120 INPUT T
130 PRINT "INPUT VALUE OF D";
140 INPUT D
150 DEF FNA(T)=SIN(T)+T^2
160 LET V=(FNA(T+D/2)-FNA(T-D/2))/D
170 PRINT "AVERAGE VELOCITY IS ";V
180 PRINT
190 GOTO 110
999 END

```

```

INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?.5
AVERAGE VELOCITY IS 2.53469

INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?.1
AVERAGE VELOCITY IS 2.54008

INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?.05
AVERAGE VELOCITY IS 2.54024

INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?.01
AVERAGE VELOCITY IS 2.5403

INPUT VALUE OF T DESIRED?2
INPUT VALUE OF D?.1
AVERAGE VELOCITY IS 3.58403

INPUT VALUE OF T DESIRED?2
INPUT VALUE OF D?.05
AVERAGE VELOCITY IS 3.58391

INPUT VALUE OF T DESIRED?2
INPUT VALUE OF D?.01
AVERAGE VELOCITY IS 3.584

```


Exercise 2 — Applications

This exercise is based upon Exercise 1, but a very important point is developed here. That is, with the computer we can handle cases that would be very difficult even if the student had completed a course in calculus. Assuming the student was acquainted with differentiation he would find great difficulty with (c). However it is handled in the same manner as (a) and (b).

```
150 DEF FNA(T)= 3*T+3-4*T+2+5
RUN
```

```
INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?1
AVERAGE VELOCITY IS 1.75
```

```
INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?.1
AVERAGE VELOCITY IS 1.00751
```

```
INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?.01
AVERAGE VELOCITY IS 1.00017
```

```
INPUT VALUE OF T DESIRED?2
INPUT VALUE OF D?.1
AVERAGE VELOCITY IS 20.0075
```

```
INPUT VALUE OF T DESIRED?2
INPUT VALUE OF D?.01
AVERAGE VELOCITY IS 20.0001
```

```
150 DEF FNA(T) = EXP(T)+T
RUN
```

```
INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?.1
AVERAGE VELOCITY IS 3.71942
```

```
INPUT VALUE OF T DESIRED?1
INPUT VALUE OF D?.01
AVERAGE VELOCITY IS 3.71833
```

```
INPUT VALUE OF T DESIRED?2
INPUT VALUE OF D?.1
AVERAGE VELOCITY IS 8.39216
```

```
INPUT VALUE OF T DESIRED?2
INPUT VALUE OF D?.01
AVERAGE VELOCITY IS 8.38928
```

```
150 DEF FNA(T) = COS(T+3)-EXP(COS(T))  
RUN
```

```
INPUT VALUE OF T DESIRED?1  
INPUT VALUE OF D?1  
AVERAGE VELOCITY IS -.633302
```

```
INPUT VALUE OF T DESIRED?1  
INPUT VALUE OF D?.1  
AVERAGE VELOCITY IS -1.08582
```

```
INPUT VALUE OF T DESIRED?1  
INPUT VALUE OF D?.01  
AVERAGE VELOCITY IS -1.08004
```

```
INPUT VALUE OF T DESIRED?2  
INPUT VALUE OF D?.1  
AVERAGE VELOCITY IS -10.5488
```

```
INPUT VALUE OF T DESIRED?2  
INPUT VALUE OF D?.01  
AVERAGE VELOCITY IS -11.2654
```

```
INPUT VALUE OF T DESIRED?2  
INPUT VALUE OF D?.0075  
AVERAGE VELOCITY IS -11.2685
```

Exercise 3 — Automatic Accuracy

This is an exercise in programming skill. The old value of the velocity is set equal to zero in line 140. The new value is computed in line 170. The test that is the heart of the process in line 180 gives the exit condition. Other than this, the program is straightforward.

```
100 REM AUTOMATIC VELOCITY PROGRAM
110 READ N
120 DEF FNA(T)=T^2
130 FOR I=1 TO N
140 LET V0=0
150 LET D=1
160 READ T
170 LET V1=(FNA(T+D/2)-FNA(T-D/2))/D
180 IF ABS((V1-V0)/V1)<.00005 THEN 220
190 LET D=D/10
200 LET V0=V1
210 GOTO 170
220 PRINT "AT T = "T" V = "V1
230 NEXT I
800 DATA 4
801 DATA 1,2,3,4
999 END
```

RUN

AT T =	1	V =	2.
AT T =	2	V =	4.
AT T =	3	V =	6.00002
AT T =	4	V =	8.00001

Exercise 4 — Advanced — Plotting Results

Hopefully, the student will discover in this exercise that for a displacement that is a linear function of time, the velocity is a constant. You might assign the students other linear functions and let them discover the property of constant velocity.

```

120 DEF FNA(T) = T
RUN

AT T = 1      V = .999999
AT T = 2      V = 1.
AT T = 3      V = 1.
AT T = 4      V = 1.

DONE

```

Exercise 5 — Rates and Trigonometric Functions

The student should discover that the desired function is the Cosine function. This is facilitated by plotting both the function for displacement and for velocity on the same graph.

```

100 REM AUTOMATIC VELOCITY PROGRAM
105 PRINT
106 PRINT " T","V1","SIN(T)","COS(T)"
107 PRINT
120 DEF FNA(T)=SIN(T)
130 FOR T=0 TO 1 STEP .1
140 LET V0=0
150 LET D=1
170 LET V1=(FNA(T+D/2)-FNA(T-D/2))/D
180 IF ABS(V1-V0)<.0001 THEN 220
190 LET D=D/10
200 LET V0=V1
210 GOTO 170
220 PRINT T,V1,SIN(T),COS(T)
230 NEXT T
999 END

```

RUN

T	V1	SIN(T)	COS(T)
0	1	0	1.
.1	.99498	9.98334E-02	.995004
.2	.980079	.198669	.980067
.3	.955343	.29552	.955336
.4	.92113	.389418	.921061
.5	.877619	.479426	.877582
.6	.825286	.564643	.825336
.7	.764847	.644218	.764842
.8	.696659	.717356	.696707
.9	.621557	.783327	.62161
1	.540257	.841471	.540302

DONE

FROM RATES TO DISPLACEMENTS

Exercise 6 — Displacement

This exercise introduces the student to the solution of differential equations. Of course, he won't know what a differential equation is; but due to the way the problem is phrased, the solution can be obtained without difficulty.

```

100  REM PROGRAM TO FIND DISPLACEMENT
110  REM GIVEN VELOCITY
120  PRINT "INPUT INITIAL VALUE OF X  ";
130  INPUT X0
140  PRINT "INPUT TIME INCREMENT  ";
150  INPUT D
160  DEF FNA(T)=COS(T)
170  FOR T=0 TO 1 STEP D
180  PRINT T,X0
190  LET X1=X0+FNA(T)*D
200  LET X0=X1
210  NEXT T
999  END

```

RUN

```

INPUT INITIAL VALUE OF X  ?0
INPUT TIME INCREMENT  ?.1
0          0
.1         .1
.2         .1995
.3         .297507
.4         .393041
.5         .485147
.6         .572905
.7         .655439
.8         .731923
.9         .801594
1          .863755

```

DONE

Exercise 7 — Increasing the Accuracy

The notion of a function is illustrated very nicely in this exercise. With the definition of the cosine function in line 160, the average velocity can be clearly defined in line 190. The student should be aware of the fact that different values of displacement are obtained using the average velocity, and that these values are more accurate.

```
100  REM PROGRAM TO FIND DISPLACEMENT
110  REM GIVEN VELOCITY
120  PRINT "INPUT INITIAL VALUE OF X ";
130  INPUT X0
140  PRINT "INPUT TIME INCREMENT ";
150  INPUT D
160  DEF FNA(T)=COS(T)
170  FOR T=0 TO 1 STEP D
180  PRINT T,X0
190  LET X1=X0+((FNA(T+D)+FNA(T))/2)*D
200  LET X0=X1
210  NEXT T
999  END
```

RUN

```
INPUT INITIAL VALUE OF X ?0
INPUT TIME INCREMENT ?.1
0          0
.1         9.97502E-02
.2         .198504
.3         .295274
.4         .389094
.5         .479026
.6         .564172
.7         .643681
.8         .716758
.9         .782674
1          .84077
```

DONE

Exercise 8 — Finding Displacement From Acceleration

This exercise needs to be related to the physics text to be most meaningful. The idea here is very powerful, and the students should work at it until the strategy is clear. The program is clear and needs no special explanation.

```

100  REM PROGRAM TO FIND DISPLACEMENT
110  REM GIVEN ACCELERATION
120  PRINT "INPUT INITIAL VALUE OF X ";
130  INPUT X0
140  PRINT "INPUT INITIAL VALUE OF V ";
150  INPUT V0
160  PRINT "INPUT TIME INCREMENT ";
170  INPUT D
180  DEF FNA(T)=T
190  FOR T=0 TO 1 STEP D
200  PRINT T,V0,X0
210  LET V1=V0+FNA(T)*D
220  LET X1=X0+V0*D
230  LET V0=V1
240  LET X0=X1
250  NEXT T
999  END

```

RUN

```

INPUT INITIAL VALUE OF X ?0
INPUT INITIAL VALUE OF V ?0
INPUT TIME INCREMENT ? .1
0          0          0
.1         0          0
.2         .01        0
.3         .03        .001
.4         .06        .004
.5         .1         .01
.6         .15        .02
.7         .21        .035
.8         .28        .056
.9         .36        .084
1          .45        .12

```

DONE

Exercise 9 — Advanced — Average Values of Velocity and Acceleration

The only change in the program is to use average values of acceleration and velocity. This is done in lines 210 and 220. This is a good program to use in a heuristic sense. Supply the student with a wide variety of acceleration functions with initial conditions on velocity and position. Then the computer can be used to great advantage to investigate the characteristics of the motion. The student should be made aware that with the computer nearly any type of motion can be studied whereas only a few types of motion can be described if only algebra is used.

```

100  REM PROGRAM TO FIND DISPLACEMENT
110  REM GIVEN ACCELERATION
120  PRINT "INPUT INITIAL VALUE OF X ";
130  INPUT X0
140  PRINT "INPUT INITIAL VALUE OF V ";
150  INPUT V0
160  PRINT "INPUT TIME INCREMENT ";
170  INPUT D
180  DEF FNA(T)=T
190  FOR T=0 TO 1 STEP D
200  PRINT T,V0,X0
210  LET V1=V0+((FNA(T+D)+FNA(T))/2)*D
220  LET X1=X0+((V1+V0)/2)*D
230  LET V0=V1
240  LET X0=X1
250  NEXT T
999  END

```

RUN

```

INPUT INITIAL VALUE OF X  ?0
INPUT INITIAL VALUE OF V  ?0
INPUT TIME INCREMENT    ?.1

```

0	0	0
.1	.005	.00025
.2	.02	.0015
.3	.045	.00475
.4	.08	.011
.5	.125	.02125
.6	.18	.0365
.7	.245	.05775
.8	.32	.086
.9	.405	.12225
1	.5	.1675

DONE

Exercise 10 — Advanced — A Generalization

The only change required is in line 190 where the FOR statement is modified.

```
190 FOR T = 0 TO 4 STEP D
RUN
```

INPUT	INITIAL VALUE OF X	?0
INPUT	INITIAL VALUE OF V	?0
INPUT	TIME INCREMENT	? .1
0	0	0
.1	.005	.00025
.2	.02	.0015
.3	.045	.00475
.4	.08	.011
.5	.125	.02125
.6	.18	.0365
.7	.245	.05775
.8	.32	.086
.9	.405	.12225
1	.5	.1675
1.1	.605	.22275
1.2	.72	.289
1.3	.845	.36725
1.4	.98	.4585
1.5	1.125	.56375
1.6	1.28	.684
1.7	1.445	.82025
1.8	1.62	.9735
1.9	1.805	1.14475
2.	2.	1.335
2.1	2.205	1.54525
2.2	2.42	1.7765
2.3	2.645	2.02975
2.4	2.88	2.306
2.5	3.125	2.60625
2.6	3.38	2.9315
2.7	3.645	3.28275
2.8	3.92	3.661
2.9	4.205	4.06725
3.	4.5	4.5025

3.1	4.805	4.96775
3.2	5.12	5.464
3.3	5.445	5.99225
3.4	5.78	6.5535
3.5	6.125	7.14875
3.6	6.48	7.779
3.7	6.845	8.44525
3.8	7.22	9.1485
3.9	7.605	9.88975
4.	8.	10.67

DONE

NEWTON'S SECOND LAW

Exercise 11 — Force and Motion

This exercise is designed to promote qualitative thinking about motion problems. It may help to have the student plot the results.

```
800 DATA 2,-4,2
RUN
```

T	V	X
0	-4	2
1	-3.	-1.55
2.	-2.	-4.1
3.	-1.	-5.65
4.	-2.83122E-06	-6.20001

DONE

```
800 DATA 2,-4,1
130 LET F = 8
RUN
```

T	V	X
0	-4	2
1	4.	1.6
2.	12.	9.2
3.	20.	24.8
4.	28.	48.4

DONE

Exercise 12 — Converging on Accuracy

As the time increment in the problem decreases the student should begin to suspect that the velocity and position will be more accurate. You can run this program with some other values for force, mass, and initial conditions to give more experience.

```
801 DATA .1,10,5
RUN
```

T	V	X
0	-4	2
1	-2.	-1.1
2.	-1.84774E-06	-2.2
3.	2.	-1.3
4.	4.	1.59999

DONE

```
801 DATA .05,20,5
RUN
```

T	V	X
0	-4	2
1.	-2.	-1.05
2.	-2.83122E-06	-2.1
3.	2.	-1.15001
4.00001	3.99999	1.79999

DONE

801 DATA .01,100,5
RUN

T	V	X
0	-4	2
.999999	-2.	-1.01
2.	-3.52412E-06	-2.02
3.00002	1.99999	-1.03001
4.00004	3.99999	1.95999

DONE

801 DATA .005,200,5
RUN

T	V	X
0	-4	2
.999999	-1.99995	-1.00498
2.00002	4.37759E-05	-2.00993
3.00004	2.00004	-1.01489
4.00007	4.00009	1.98017

DONE

Exercise 13 — A Variable Force

A simple change in the program is required. Now the force is not a constant and must be computed each time step. This is done in line 185 in the program. It will be very easy for the student to respond to this simple change. You may wish to point out the vast difference in the analytical techniques produced by such a simple change. It is precisely at points such as this that the power of the computer becomes visible.

```

100 REM NEWTON'S SECOND LAW
110 READ X0,V0,M
120 READ D,N,L
140 PRINT
150 PRINT "T","V","X"
160 PRINT
170 LET C=N
180 FOR T=0 TO L STEP D
185 LET F=4*COS(T)
190 IF C<N THEN 220
200 PRINT T,V0,X0
210 LET C=0
220 LET V1=V0+F*D/M
230 LET X1=X0+V0*D
240 LET V0=V1
250 LET X0=X1
260 LET C=C+1
270 NEXT T
800 DATA 2,-4,1
801 DATA .1,10,5
999 END

```

RUN

T	V	X
0	-4	2
1	-.544982	-.305178
2.	-8.26126E-02	-.322429
3.	-3.03799	-1.52959
4.	-6.69396	-6.31051

DONE

Exercise 14 — Advanced — A Challenge

This problem would be very difficult by analytical methods. However with the computer it merely requires several tests to locate the current value of displacement which in turn determines the appropriate value of force to use. The program follows without difficulty.

```
100 REM NEWTON'S SECOND LAW
110 READ X0,V0,M
120 READ D,N,L
140 PRINT
150 PRINT "T","V","X"
160 PRINT
170 LET C=N
180 FOR T=0 TO L STEP D
181 IF ABS(X0)<5 THEN 185
182 IF X0 >= 5 THEN 187
183 LET F=+4
184 GOTO 190
185 LET F=0
186 GOTO 190
187 LET F=-4
190 IF C<N THEN 220
200 PRINT T,V0,X0
210 LET C=0
220 LET V1=V0+F*D/M
230 LET X1=X0+V0*D
240 LET V0=V1
250 LET X0=X1
260 LET C=C+1
270 NEXT T
800 DATA 0,5,1
801 DATA .1,5,10
999 END
```


RUN

T	V	X
0	5	0
.5	5	2.5
1	5	5
1.5	3.	7.1
2.	1.	8.2
2.5	-.999999	8.3
3.	-3.	7.4
3.5	-5.	5.5
4.	-5.8	2.72
4.5	-5.8	-.179999
5.	-5.8	-3.08
5.5	-5.4	-5.98
6.	-3.4	-8.28
6.50001	-1.4	-9.58
7.00001	.6	-9.88
7.50001	2.6	-9.18
8.00001	4.6	-7.48
8.50001	6.6	-4.78
9.00002	6.6	-1.48
9.50002	6.6	1.82

DONE

Exercise 15 — Advanced — Barrier Penetration

The only difficult point in this exercise is to insure that each step produces a positive velocity as long as x is less than 5. A statement must be provided to test for this condition. If the test is failed the object is moving in the negative direction and will not penetrate. For your better students, this problem is a good one for which to develop an automatic program to replace the trial and error approach.

```
100  REM BARRIER SIMULATION
110  INPUT V0
115  LET A=V0
120  LET X0=0
130  LET V1=V0-5.00000E-02
140  LET X1=X0+1.00000E-02*V0
145  IF X1>X0 THEN 150
146  PRINT "VELOCITY TOO LOW"
147  GOTO 110
150  LET V0=V1
160  LET X0=X1
170  IF X0<5 THEN 130
180  PRINT
190  PRINT A,V0
195  PRINT
200  GOTO 110
999  END
```

RUN

?10

10	7.05003
----	---------

?5

VELOCITY TOO LOW

?7.5

7.5	2.49998
-----	---------

?7

VELOCITY TOO LOW

?7.25

7.25	1.64998
------	---------

?7.125

7.125	1.02498
-------	---------

?7.05

7.05	.199983
------	---------

?7.04

VELOCITY TOO LOW

?7.045

VELOCITY TOO LOW

?7.0475

7.0475	.097482
--------	---------

?

DONE

HALF STEP METHOD

Exercise 16 — Half Step Computation

The half step is very easy to insert. This is done in line 175 in the program. Note that this does throw the velocity out of step. However the increase in accuracy for the position is dramatic. The values of position in the printout are exact with the exception of the last value. The exact value for $t = 4$ is 2.

```

100  REM NEWTON'S SECOND LAW
110  READ X0,V0,M
120  READ D,N,L
130  LET F=2
140  PRINT
150  PRINT "T","V","X"
160  PRINT
170  LET C=N
175  LET V0=V0+(F/M)*D/2
180  FOR T=0 TO L STEP D
190  IF C<N THEN 220
200  PRINT T,V0,X0
210  LET C=0
220  LET V1=V0+F*D/M
230  LET X1=X0+V0*D
240  LET V0=V1
250  LET X0=X1
260  LET C=C+1
270  NEXT T
800  DATA 2,-4,1
801  DATA .1,10,5
999  END

```

RUN

T	V	X
0	-3.9	2
1	-1.9	-1.
2.	9.99982E-02	-2.
3.	2.1	-1.
4.	4.1	1.99999

DONE

Exercise 17 — Half Step Computation

This exercise uses the same program developed for Exercise 16 with the data used in Exercise 12.

801 DATA .1,10,5
RUN

T	V	X
0	-3.9	2
1	-1.9	-1.
2.	9.99982E-02	-2.
3.	2.1	-1.
4.	4.1	1.99999

DONE

801 DATA .05,20,5
RUN

T	V	X
0	-3.95	2
1.	-1.95	-1.
2.	4.99974E-02	-2.
3.	2.05	-1.00001
4.00001	4.05	1.99999

DONE

801 DATA .01,100,5
RUN

T	V	X
0	-3.99	2
.999999	-1.99	-1.
2.	9.99671E-03	-2.
3.00002	2.01	-1.00001
4.00004	4.00999	1.99999

DONE

801 DATA .005,200,5
RUN

T	V	X
0	-3.995	2
.999999	-1.99495	-.999977
2.00002	5.04391E-03	-1.99993
3.00004	2.00504	-.999889
4.00007	4.00509	2.00018

DONE

THE HARMONIC OSCILLATOR

Exercise 18 — Harmonic Oscillator

The program for this exercise poses no special difficulty. When the results are available, see if the students can see what kind of mathematical function would describe them. Generally, if the results are plotted the students will see the Sine and Cosine curves. This leads naturally to Exercise 19.

```
100  REM NEWTON'S SECOND LAW
110  READ X0,V0,M
120  READ N
130  LET P=3.14159
132  LET D=P/100
140  PRINT
150  PRINT "T","V","X"
160  PRINT
170  LET C=N
180  FOR T=0 TO 2*P STEP P/100
185  LET F=-X0
190  IF C<N THEN 220
200  PRINT T,V0,X0
210  LET C=0
220  LET V1=V0+F*D/M
230  LET X1=X0+V0*D
240  LET V0=V1
250  LET X0=X1
260  LET C=C+1
270  NEXT T
800  DATA 1,0,1
801  DATA 10
999  END
```

RUN

T	V	X
0	0	1
.314159	-.310446	.955791
.628318	-.593443	.81716
.942477	-.820892	.596802
1.25664	-.969876	.315576
1.5708	-1.02497	5.30496E-04
1.88496	-.979821	-.317691
2.19911	-.837879	-.607827
2.51327	-.61214	-.841072
2.82743	-.32397	-.993926
3.14159	-1.08758E-03	-1.05056
3.45575	.325103	-1.00445
3.76991	.622559	-.859122
4.08407	.861748	-.62787
4.39823	1.01857	-.332587
4.71239	1.07679	-1.67247E-03
5.02655	1.02971	.332687
5.34071	.880903	.637648
5.65487	.644004	.882931
5.96903	.341431	1.04383

DONE

Exercise 19 — Half Step Computation

The half step computation is in line 175.

```
100 REM NEWTON'S SECOND LAW
110 READ X0,V0,M
120 READ N
130 LET P=3.14159
132 LET D=P/100
140 PRINT
150 PRINT "T","V","X"
160 PRINT
170 LET C=N
175 LET V0=V0-(X0/M)*(D/2)
180 FOR T=0 TO 2*P STEP P/100
185 LET F=-X0
190 IF C<N THEN 220
200 PRINT T,V0,X0
210 LET C=0
220 LET V1=V0+F*D/M
230 LET X1=X0+V0*D
240 LET V0=V1
250 LET X0=X1
260 LET C=C+1
270 NEXT T
800 DATA 1,0,1
801 DATA 10
999 END
```


RUN

T	V	X
0	-.015708	1
.314159	-.32546	.950915
.628318	-.606279	.807838
.942477	-.830266	.583908
1.25664	-.974833	.300341
1.5708	-1.02498	-1.55698E-02
1.88496	-.97483	-.333082
2.19911	-.828331	-.620989
2.51327	-.598928	-.850687
2.82743	-.308357	-.999014
3.14159	1.54147E-02	-1.05058
3.45575	.340881	-.999347
3.76991	.636054	-.849342
4.08407	.87161	-.614334
4.39823	1.02379	-.316587
4.71239	1.07682	1.52419E-02
5.02655	1.02448	.348862
5.34071	.870886	.651485
5.65487	.630135	.893047
5.96903	.325034	1.04919

DONE

Exercise 20 — Changing the Mass

This exercise should be correlated with the physics textbook for best results. Also, the student should plot the results to assist him to better observe the effects of the change in mass.

800 DATA 1,0,5
RUN

T	V	X
0	-3.14159E-03	1
.314159	-6.57968E-02	.990141
.628318	-.127278	.960732
.942477	-.18637	.912315
1.25664	-.241903	.845808
1.5708	-.292774	.762486
1.88496	-.33797	.66396
2.19911	-.37659	.552142
2.51327	-.407858	.429213
2.82743	-.431143	.297575
3.14159	-.44597	.159807
3.45575	-.452029	1.86155E-02
3.76991	-.449185	-.123223
4.08407	-.437475	-.26291
4.39823	-.417112	-.397688
4.71239	-.388482	-.524888
5.02655	-.352133	-.641987
5.34071	-.308766	-.746653
5.65487	-.259224	-.836799
5.96903	-.20447	-.910618

DONE

800 DATA 1,0,2
RUN

T	V	X
0	-7.85398E-03	1
.314159	-.16383	.975392
.628318	-.312508	.902617
.942477	-.446529	.785023
1.25664	-.559217	.628179
1.5708	-.644918	.439618
1.88496	-.699276	.22847
2.19911	-.719458	5.02351E-03
2.51327	-.704303	-.219775
2.82743	-.654381	-.43486
3.14159	-.571979	-.629589
3.45575	-.460997	-.79427
3.76991	-.326758	-.920642
4.08407	-.17576	-1.00229
4.39823	-1.53575E-02	-1.03498
4.71239	.146594	-1.01684
5.02655	.302127	-.948531
5.34071	.443546	-.833161
5.65487	.563818	-.676183
5.96903	.656912	-.485121

DONE

Exercise 21 — Advanced — The Pendulum

The goal in this exercise is to encourage the student to derive the equations below which describe the motion of a simple pendulum of length L .

$$\Delta\omega/\Delta t = - (g/L)\theta$$

$$\Delta\theta/\Delta t = \omega$$

Then, the solution can be obtained from the harmonic oscillator program.

NOTES

MORE COMPLICATED FORCES

Exercise 22 — Damped Harmonic Oscillator

The only change required is in the line specifying the force.

```
100 REM NEWTON'S SECOND LAW
110 READ X0,V0,M
120 READ N
130 LET P=3.14159
132 LET D=P/100
140 PRINT
150 PRINT "T","V","X"
160 PRINT
170 LET C=N
175 LET V0=V0-(X0/M)*(D/2)-(V0/M)*(D/2)
180 FOR T=0 TO 2*P STEP P/100
185 LET F=-X0-V0
190 IF C<N THEN 220
200 PRINT T,V0,X0
210 LET C=0
220 LET V1=V0+F*D/M
230 LET X1=X0+V0*D
240 LET V0=V1
250 LET X0=X1
260 LET C=C+1
270 NEXT T
800 DATA 1,0,1
801 DATA 10
999 END
```

RUN

T	V	X
0	-.015708	1
.314159	-.280771	.95506
.628318	-.451363	.840397
.942477	-.53801	.684351
1.25664	-.555616	.511266
1.5708	-.521031	.340473
1.88496	-.451085	.185967
2.19911	-.361159	5.66313E-02
2.51327	-.264255	-4.31658E-02
2.82743	-.170514	-.112743
3.14159	-8.71108E-02	-.154183
3.45575	-1.84299E-02	-.171423
3.76991	3.35695E-02	-.169421
4.08407	6.88752E-02	-.153463
4.39823	.088906	-.128625
4.71239	9.60094E-02	-9.93906E-02
5.02655	9.30146E-02	-6.94284E-02
5.34071	.082862	-.041494
5.65487	6.83225E-02	-1.74373E-02
5.96903	5.18057E-02	1.71561E-03

DONE

Exercise 23 — An Experiment

This is very good combination laboratory and computer exercise.

Exercise 24 — Advanced — Driven, Damped Harmonic Oscillator

As in Exercise 22, all that is required is modification of the forces.

```

100 REM NEWTON'S SECOND LAW
110 READ X0,V0,M
120 READ N
130 LET P=3.14159
132 LET D=P/100
134 LET T=0
140 PRINT
150 PRINT "T","V","X"
160 PRINT
170 LET C=N
175 LET V0=V0-(X0/M)*(D/2)-(V0/M)*(D/2)+(2*COS(T)/M)*(D/2)
180 FOR T=0 TO 2*P STEP P/100
185 LET F=-X0-V0+2*COS(T)
190 IF C<N THEN 220
200 PRINT T,V0,X0
210 LET C=0
220 LET V1=V0+F*D/M
230 LET X1=X0+V0*D
240 LET V0=V1
250 LET X0=X1
260 LET C=C+1
270 NEXT T
800 DATA 1,0,1
801 DATA 10
999 END

```


RUN

T	V	X
0	.015708	1
.314159	.272497	1.04444
.628318	.386542	1.15026
.942477	.338567	1.26912
1.25664	.13793	1.35074
1.5708	-.182084	1.35134
1.88496	-.571079	1.24018
2.19911	-.969553	1.00389
2.51327	-1.31718	.64809
2.82743	-1.56052	.196585
3.14159	-1.65937	-.311768
3.45575	-1.59135	-.827912
3.76991	-1.35415	-1.29856
4.08407	-.965564	-1.67259
4.39823	-.461242	-1.90704
4.71239	.109376	-1.97227
5.02655	.688574	-1.85547
5.34071	1.21664	-1.56242
5.65487	1.63829	-1.11719
5.96903	1.9085	-.559911

DONE

ORBITAL MOTION

Exercise 25 — Orbital Motion

This problem should cause no particular difficulty. The symbol assignment must be given particular attention however. The solution given contains a half step computation in the velocity.

```

100 REM ORBITAL PROBLEM
110 READ X1,X2,V1,V2
111 READ D,N,L
120 LET R1=(X12+X22)*1.5
121 LET V1=V1-(X1*D)/(R1*2)
122 LET V2=V2-(X2*D)/(R1*2)
125 LET C=N
130 FOR T=0 TO L STEP D
135 IF C<N THEN 151
140 PRINT T,X1,X2,R1
141 LET C=0
151 LET X1=X1+V1*D
152 LET X2=X2+V2*D
153 LET R1=(X12+X22)*1.5
154 LET V1=V1-X1*D/R1
155 LET V2=V2-X2*D/R1
156 LET C=C+1
157 NEXT T
800 DATA 1,0,0,1
801 DATA 1.00000E-02,10,1
999 END
  
```

RUN

0	1	0	1.
.1	.995004	9.98351E-02	1.
.2	.980066	.198673	1.
.3	.955336	.295525	1.
.4	.921061	.389425	1.00001
.5	.877582	.479433	1.00001
.6	.825335	.564652	1.00001
.7	.764841	.644228	1.00002
.799999	.696706	.717368	1.00002
.899999	.621609	.78334	1.00003
.999999	.540301	.841485	1.00003

DONE

Exercise 26 — Experimentation

This is strictly a *discovery* exercise. Students should be encouraged to try a wide range of initial positions and velocities and plot the resultant orbits. See if they can predict the orbit that will result from given initial conditions.

Exercise 27 — A Challenge

The initial conditions place the object at rest under the influence of a central force. As indicated in the printout, the object begins to acquire velocity towards the center of force as would be expected. The unexpected part is how can the object arrive at a point further out on the other side than its initial distance? The problem is caused by the large changes in velocity that take place as the object nears the center of force. These large changes invalidate the whole solution. Students must constantly be on the alert for this type of error.

```
800 DATA 1,0,0,0
```

```
801 DATA .01,10,1.5
```

```
RUN
```

0	1	0	1.
.1	.994992	0	.985051
.2	.979865	0	.940804
.3	.954303	0	.869078
.4	.917733	0	.772946
.5	.869252	0	.656805
.6	.807476	0	.526488
.7	.730258	0	.389429
.799999	.634073	0	.254929
.899999	.512477	0	.134593
.999999	.350752	0	4.31519E-02
1.1	8.03883E-02	0	5.19491E-04
1.2	-2.13429	0	9.72217
1.3	-4.52498	0	92.6509
1.4	-6.91509	0	330.669
1.5	-9.30498	0	805.65

```
DONE
```

Exercise 28 — Advanced — A Double Force Center

The solution is obtained by adding an additional force to that considered previously. One of the fixed masses is located at (W,O) with the other at (-W,O). This program is fun to play with. Students can discover very interesting orbits and effects.

```
100 REM DOUBLE FORCE CENTER
110 READ X1,X2,V1,V2
115 READ D,N,L
116 READ W
120 LET R1=((X1-W)^2+X2^2)^.5
130 LET R2=((X1+W)^2+X2^2)^.5
140 LET V1=V1-(X1-W)*D/(R1^2)-(X1+W)*D/(R2^2)
150 LET V2=V2-(X2*D)/(R1^2)-(X2*D)/(R2^2)
160 LET C=N
170 FOR T=0 TO L STEP D
180 IF C<N THEN 210
190 PRINT T,X1,X2
200 LET C=0
210 LET X1=X1+V1*D
220 LET X2=X2+V2*D
230 LET R1=((X1-W)^2+X2^2)^.5
240 LET R2=((X1+W)^2+X2^2)^.5
250 LET V1=V1-(X1-W)*D/R1-(X1+W)*D/R2
260 LET V2=V2-(X2*D)/R1-(X2*D)/R2
270 LET C=C+1
280 NEXT T
800 DATA 1.5,0,0,1.5
801 DATA 1.00000E-02,10,2
802 DATA 1
999 END
```

RUN

0	1.5	0
.1	1.47937	.148018
.2	1.41946	.28435
.3	1.32587	.398869
.4	1.20676	.48417
.5	1.07155	.536144
.6	.929619	.553918
.7	.789188	.539354
.799999	.656693	.496284
.899999	.536439	.429703
.999999	.430599	.345035
1.1	.339437	.247546
1.2	.261709	.141963
1.3	.195149	3.22819E-02
1.4	.136995	-7.82628E-02
1.5	8.44511E-02	-.187123
1.6	3.50322E-02	-.292324
1.7	-1.32462E-02	-.392354
1.8	-6.18257E-02	-.486073
1.9	-.111659	-.572639
2.	-.163308	-.65144

DONE

Scan Copyright ©
The Museum of HP Calculators
www.hpmuseum.org

Original content used with permission.

Thank you for supporting the Museum of HP
Calculators by purchasing this Scan!

Please to not make copies of this scan or
make it available on file sharing services.